Integrace speciálních funkcí

Racionální funkce

Integraci racionálních funkcí zvládá *Mathematica* většinou dobře, ale jsou i jednoduché příklady, s kterými má program potíže. Nepotřebujeme sice k integraci znát rozklad racionálních funkcí na částečné zlomky, ale někdy je pro kontrolu vhodné takovýto rozklad provést. Používá se příkaz Apart. Někdy je výsledek uveden ve zvláštním tvaru a je pak vhodné použít Simplify nebo Expand apod. pro vhodnější tvar.

```
Apart[(x + 4) / (x^3 + 3 x^2 - 10 x), x]
```

$$-\frac{2}{5x}-\frac{1}{35(x+5)}+\frac{3}{7(x-2)}$$

```
Integrate[(x + 4) / (x^3 + 3 x^2 - 10 x), x]
```

$$\frac{3}{7}\log(2-x)-\frac{2\log(x)}{5}-\frac{1}{35}\log(x+5)$$

```
Apart[(x^5 + x^4 - 1) / (x^3 + x + 2), x]
.
```

$$x^2+\frac{6-11x}{4\left(x^2-x+2\right)}+x-\frac{1}{4(x+1)}-1$$

```
Integrate[(x^5 + x^4 - 1) / (x^3 + x + 2), x]
```

$$\frac{1}{24}\left(4x\left(2x^2+3x-6\right)-33\log\left(x^2-x+2\right)-6\log(x+1)\right)+\frac{\tan^{-1}\left(\frac{2x-1}{\sqrt{7}}\right)}{4\sqrt{7}}$$

```
Expand[%]
```

$$\frac{x^3}{3}+\frac{x^2}{2}-\frac{11}{8}\log\left(x^2-x+2\right)-x-\frac{1}{4}\log(x+1)+\frac{\tan^{-1}\left(\frac{2x-1}{\sqrt{7}}\right)}{4\sqrt{7}}$$

```
Apart[1/((x^3 + 1) (x^2 + 1)^3), x]
```

$$\frac{x-2}{3\left(x^2-x+1\right)}-\frac{3\,(x-1)}{8\left(x^2+1\right)}+\frac{x+3}{4\left(x^2+1\right)^2}+\frac{x+1}{2\left(x^2+1\right)^3}+\frac{1}{24\,(x+1)}$$

```
Integrate[1/((x^3 + 1) (x^2 + 1)^3), x]
```

$$\frac{1}{48}\left(6\log\left(x^3+1\right)+\frac{6\,(x-1)}{\left(x^2+1\right)^2}+\frac{3\,(9\,x-2)}{x^2+1}-9\log\left(x^2+1\right)+2\log\left(x^2-x+1\right)-4\log(x+1)+45\tan^{-1}(x)-16\sqrt{3}\,\tan^{-1}\left(\frac{2\,x-1}{\sqrt{3}}\right)\right)$$

Je záhadou, že *Mathematica* neumí rozložit funkci $1/(x^4+1)$, ale umí rozložit funkci $1/(x^4+4)$. Zintegrovat tyto funkce umí.

```
Apart[1/(x^4 + 1), x]
```

$$\frac{1}{x^4+1}$$

```
Apart[1/(x^4 + 4), x]
```

$$\frac{2-x}{8\left(x^2-2\,x+2\right)}+\frac{x+2}{8\left(x^2+2\,x+2\right)}$$

```
Integrate[1/(x^4 + 1), x]
```

$$\frac{-\log\left(x^2-\sqrt{2}\,x+1\right)+\log\left(x^2+\sqrt{2}\,x+1\right)-2\tan^{-1}\left(1-\sqrt{2}\,x\right)+2\tan^{-1}\left(\sqrt{2}\,x+1\right)}{4\sqrt{2}}$$

```
Integrate[1/(x^4 + 4), x]
```

$$-\frac{1}{16}\log\left(x^2-2\,x+2\right)+\frac{1}{16}\log\left(x^2+2\,x+2\right)-\frac{1}{8}\tan^{-1}(1-x)+\frac{1}{8}\tan^{-1}(x+1)$$

Chceme-li si osvěžit integrály z jednoduchých racionálních funkcí, musíme mít na paměti varování z předchozí kapitoly, že *Mathematica* nedává předpoklady, za kterých výsledek platí. V prvním příkladě platí výsledek pro n≠1, ve třetím by měl být výraz pod odmocnonou kladný, pro záporný výraz se dostane jiný výsledek a nepomůže ani dodání předpokladu, ani počítání jen v reálných číslech. Konkrétní příklady však program spočte správně.

```
Integrate[1 / (a x + b) ^n, x]
```

$$\frac{(a\,x+b)^{1-n}}{a-a\,n}$$

```
Integrate[1 / (a x + b), x]
```

$$\frac{\log(a\,x+b)}{a}$$

```
Integrate[1 / (a x^2 + b x + c), x]
```

$$\frac{2\tan^{-1}\left(\frac{2\,a\,x+b}{\sqrt{4\,a\,c-b^2}}\right)}{\sqrt{4\,a\,c-b^2}}$$

```
Integrate[1 / (x^2 - 1), x]
```

$$\frac{1}{2}\log(1-x)-\frac{1}{2}\log(x+1)$$

```
Integrate[1 / (x^2 - 4 x + 1), x]
```

$$\frac{\log\left(-x+\sqrt{3}+2\right)-\log\left(x+\sqrt{3}-2\right)}{2\sqrt{3}}$$

```
Integrate[1 / (a x^2 + b x + c), x, Assumptions → {4 a c < b^2}]
```

$$\frac{2\tan^{-1}\left(\frac{2\,a\,x+b}{\sqrt{4\,a\,c-b^2}}\right)}{\sqrt{4\,a\,c-b^2}}$$

```
<< Miscellaneous`RealOnly`
```

— *General::obspkg :*

Miscellaneous`RealOnly` is now obsolete. The legacy version being loaded may conflict with current Mathematica functionality.
See the Compatibility Guide for updating information. ≫

```
Integrate[1 / (a x^2 + b x + c), x, Assumptions → {4 a c < b^2}]
```

$$\frac{2\tan^{-1}\left(\frac{2ax+b}{\sqrt{4ac-b^2}}\right)}{\sqrt{4ac-b^2}}$$

Funkce s odmocninami

Na integraci racionálních funkcí lze převést mnoho integrálů z funkcí s odmocninami. Mathematica takové integrály většinou vypočte bez našeho zásahu a není třeba příslušné substituce provádět. Někdy však může být vhodné ukázat, jak se taková funkce na racioánlí převede. Nebudeme předvádět všechny možné případy z teorie, ukážeme jeden typický příklad.

```
Integrate[1 / (x + Sqrt[x^2 + x + 1]), x]
```

$$\sqrt{x^2+x+1}-\log\left(2\sqrt{x^2+x+1}-x+1\right)-x+2\log(x+1)-\frac{1}{2}\sinh^{-1}\left(\frac{2x+1}{\sqrt{3}}\right)$$

Zapamatujeme si výsledek a dáme vhodnou substituci. Není třeba ji vypisovat, ani příslušnou derivaci, ale pro zajímavost je vypíšeme

```
Clear[f1]; f1[x_] := %
```

```
subst[t_] := x /. Solve[Sqrt[x^2 + x + 1] == -x + t, x][[1]]
```

```
subst[t]
```

$$\frac{t^2-1}{2t+1}$$

```
D[subst[t], t]
```

$$\frac{2\,t}{2\,t+1}-\frac{2\left(t^2-1\right)}{(2\,t+1)^2}$$

```
Together[%]
```

$$\frac{2\left(t^2+t+1\right)}{(2\,t+1)^2}$$

Program jen automaticky dosadí substituci a je třeba provést úpravy (zvláště upravit $\sqrt{x^2}$ na x).

```
1 / (x + Sqrt[x^2 + x + 1]) * Together[D[subst[t], t]] /.
  x → subst[t]
```

$$\frac{2\left(t^2+t+1\right)}{(2\,t+1)^2\left(\frac{t^2-1}{2\,t+1}+\sqrt{\frac{(t^2-1)^2}{(2\,t+1)^2}+\frac{t^2-1}{2\,t+1}+1}\right)}$$

```
Simplify[%]
```

$$\frac{2\left(t^2+t+1\right)}{(2\,t+1)\left(t^2+2\sqrt{\frac{(t^2+t+1)^2}{(2\,t+1)^2}}\;t+\sqrt{\frac{(t^2+t+1)^2}{(2\,t+1)^2}}-1\right)}$$

```
Simplify[%] /. Sqrt[(t^2 + t + 1)^2/(2 t + 1)^2] -> (t^2 + t + 1)/(2 t + 1)
```

$$\frac{2\left(t^2+t+1\right)}{t\,(2\,t+1)^2}$$

Do výsledku integrace dosadíme zpátky proměnnou x. Výsledek vypadá jinak než výsledek nahoře, nicméně, ověříme snadno, že jde o stejné funkce.

```
Integrate[2 (t^2 + t + 1)/(t (2 t + 1)^2), t]
```

$$2\left(\frac{3}{4\,(2\,t+1)}+\log(t)-\frac{3}{4}\log(2\,t+1)\right)$$

```
% /. t → x + Sqrt[x^2 + x + 1]
```

$$2\left(\frac{3}{4\left(2\left(\sqrt{x^2+x+1}+x\right)+1\right)}+\log\left(\sqrt{x^2+x+1}+x\right)-\frac{3}{4}\log\left(2\left(\sqrt{x^2+x+1}+x\right)+1\right)\right)$$

```
Clear[f2]; f2[x_] := %
```

```
f1[x] == f2[x]
```

True

Racionální funkce goniometrických funkcí

Mathematica někdy rozpozná, kdy je vhodnější dát substituci např. sin místo tg apod. Při substituci ttg neslepí výsledek ve spojitou funkci.

```
Integrate[1 / (1 + 3 (Cos[x]) ^2), x]
```

$$-\frac{1}{2}\tan^{-1}(2\cot(x))$$

```
Plot[%, {x, 0, 2 Pi}]
```




```
Integrate[Sin[x] Cos[x] / ((Sin[x])^4 + (Cos[x])^4), x]
```

$$-\frac{1}{2}\tan^{-1}(\cos(2x))$$

```
Plot[%, {x, 0, 2 Pi}]
```



```
Integrate[1 / (2 - Sin[x]), x]
```

$$-\frac{2\tan^{-1}\left(\frac{1-2\tan\left(\frac{x}{2}\right)}{\sqrt{3}}\right)}{\sqrt{3}}$$

```
Plot[%, {x, 0, 2 Pi}]
```



Proveďme pro zajímavost substituci u posledního příkladu. Nebudeme již komentovat příslušné kroky, postup je stejný jako u substituce odmocniny.

```
Clear[subst]; subst[t_] := x /. Solve[Tan[x / 2] == t, x][[1]]
```

```
1 / (2 - Sin[x]) * Together[D[subst[t], t]] /. x → subst[t]
```

— *Solve::ifun :*

Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information. ≫

— *Solve::ifun :*

Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information. ≫

$$\frac{2}{(t^2+1)\left(2-\sin\left(2\tan^{-1}(t)\right)\right)}$$

```
FullSimplify[%, Trig → True]
```

$$\frac{1}{(t-1)\,t+1}$$

```
Integrate[%, t]
```

$$\frac{2\tan^{-1}\left(\frac{2\,t-1}{\sqrt{3}}\right)}{\sqrt{3}}$$

```
% /. t → Tan[x / 2]
```

$$\frac{2\tan^{-1}\left(\frac{2\tan\left(\frac{x}{2}\right)-1}{\sqrt{3}}\right)}{\sqrt{3}}$$

$$\mathbf{FullSimplify}\left[-\frac{2\tan^{-1}\left(\frac{1-2\tan\left(\frac{x}{2}\right)}{\sqrt{3}}\right)}{\sqrt{3}}-\frac{2\tan^{-1}\left(\frac{2\tan\left(\frac{x}{2}\right)-1}{\sqrt{3}}\right)}{\sqrt{3}}\right]$$

$$0$$